
* Program.: ADD.PRG 
* Author..: Stephen Kurasch 
* Date....: June 10, 1985 
* Notice..: Copyright 1985, Ashton-Tate 
* Notes...: Entering deposits for checkbook management system 
*           Called from CBMENU.PRG 
 
USE Deposits 
 
* remember where we are 
firstdep = RECCOUNT() + 1 
 
* initialize logical memory variables 
STORE .F. TO again,special 
 
DO WHILE .T. 
 
   * draw deposit slip 
   CLEAR 
   @  3, 0 TO 20,45 
   @  6, 3 SAY LEFT(dhoriz,27)+CHR(209)+LEFT(dhoriz,12) 
   @  7,30 TO 14,30 
 
   @  2,18 SAY "DEPOSIT SLIP" 
   @  5, 9 SAY "RECEIVED FROM" 
   @  5,34 SAY "AMOUNT" 
   @ 16,15 SAY "SUBTOTAL" 
 
   * initialize memory variables 
   STORE 0.00 TO subtotal,totals,cash 
 
*--------------------------------------------------------------* 
*                                                              * 
*    Programming Note:  The next lines are discussed in        * 
*    Chapter 7 of "Programming in dBASE III PLUS"              * 
*                                                              * 
*--------------------------------------------------------------* 
 
   * display subtotal 
   @ 16,32 SAY subtotal PICTURE "9999999.99" 
 
   * enter date if first time through loop 
   IF .NOT. again 
      mdate = today 
 
      * get deposit date 
      DO WHILE .T. 
         @ 4,5 SAY "ENTER DATE OF DEPOSIT " GET mdate 
         READ 
 
         * correct date? 
         answer = " " 
         DO WHILE .NOT. answer$"YyNn" 
            answer = " " 



            @ 4,48 SAY "Is this correct? (Y/N)" GET answer 
            READ 
         ENDDO 
 
         * erase question from the screen 
         @ 4,48 
 
         IF UPPER(answer)="Y" 
            EXIT 
         ENDIF 
 
      ENDDO 
 
      * display date of deposit 
      @ 4,4 SAY "    Date of Deposit:             " 
      @ 4,25 SAY mdate 
 
   ENDIF 
 
   * initialize memory variables 
   line = 7 
   restart = .F. 
 
   lisa = .T. 
   DO WHILE lisa 
   * lisa is for updating incorrect deposit slip 
 
      * loop for entering and updating individual deposit entries 
      DO WHILE .T. 
 
         * when deposit slip gets too long, add another line 
         IF line > 14 
            @ 16,15 SAY SPACE(28) 
            @ 20, 0 
            @ 24, 0 
            ? 
            @  0, 0 
            @ 19, 0 SAY svert 
            @ 19,45 SAY svert 
            @ 20, 0 SAY CHR(192)+LEFT(shoriz,44)+CHR(217) 
            @ 14,30 SAY svert 
            @ 16,15 SAY "SUBTOTAL" 
            @ 16,32 SAY subtotal PICTURE "9999999.99" 
            line = 14 
         ENDIF 
 
         * if second time through loop 
         IF again 
            IF .NOT. special 
               mpayfrom = Paidfrom 
               mamt = Amt 
            ENDIF 
 
            * display deposit information 



            @ line,4 SAY mpayfrom 
            @ line,32 SAY mamt PICTURE "9999999.99" 
 
            * correct entry? 
            ans = " " 
            DO WHILE .NOT. ans$"YyNn" .AND. .NOT. special 
               ans = " " 
               @ line,48 SAY "Is this correct? (Y/N)" GET ans 
               READ 
            ENDDO 
 
            @ line,48 
 
            IF UPPER(ans)="N" .OR. special 
 
               * delete transaction? 
               an = " " 
               DO WHILE .NOT. an$"YyNn" 
                  an = " " 
                  @ line,52 SAY "Do you wish to delete" 
                  @ line+1,50 SAY "this transaction? (Y/N)" GET an 
                  READ 
               ENDDO 
 
               @ line,48 CLEAR 
 
               IF UPPER(an)="Y" 
                  * erase current deposit from screen... 
                  @ line,4 SAY SPACE(25) 
                  @ line,32 SAY SPACE(10) 
                  * ...delete record and continue 
                  DELETE 
                  SKIP 
                  IF EOF() 
                     EXIT 
                  ENDIF 
                  LOOP 
               ELSE 
                  IF special 
                     REPLACE Paidfrom WITH mpayfrom 
                     REPLACE Amt WITH mamt 
                  ENDIF 
               ENDIF 
 
            ELSE 
               subtotal = subtotal + mamt 
               @ 16,32 SAY subtotal PICTURE "9999999.99" 
               * skip to next record 
               SKIP 
               IF EOF() 
                  EXIT 
               ENDIF 
               line = line + 1 
               LOOP 



            ENDIF 
 
         ELSE 
            * initialize memory variables 
            mpayfrom = SPACE(25) 
            mamt = 0.00 
            loopamt = .F. 
         ENDIF 
 
         special = .F. 
 
         * this routine is to enter deposit data 
         @ line,4 GET mpayfrom 
         READ 
 
         * if no entry, clear out line and exit loop 
         IF mpayfrom=SPACE(25) 
            @ line,4 SAY SPACE(25) 
            @ line,32 SAY SPACE(10) 
 
            * if second time through loop, we do not want to exit 
            IF again 
               special = .T. 
               LOOP 
            ENDIF 
            EXIT 
         ENDIF 
 
         * input positive non-zero amount 
         DO WHILE .T. 
            @ line,32 GET mamt PICTURE "9999999.99" 
            READ 
            IF mamt > 0.00 
               EXIT 
            ENDIF 
            mamt = 0.00 
         ENDDO 
 
         * display subtotal to screen 
         test = subtotal + mamt 
         @ 16,32 SAY test PICTURE "9999999.99" 
 
         * correct entry? 
         ans = " " 
         DO WHILE .NOT. ans$"YyNn" 
            ans = " " 
            @ line,48 SAY "Is this correct? (Y/N)" GET ans 
            READ 
         ENDDO 
 
         @ line,48 
 
         * if not correct - display subtotal, clear entry and loop 
         IF UPPER(ans)="N" 



            @ 16,32 SAY subtotal PICTURE "9999999.99" 
            @ line,32 SAY SPACE(10) 
            LOOP 
         ENDIF 
 
         subtotal = subtotal + mamt 
 
         * now add a record (if first time through)... 
         IF .NOT. again 
            APPEND BLANK 
         ENDIF 
 
         *...and replace data 
         REPLACE Paidfrom WITH mpayfrom 
         REPLACE Date WITH mdate 
         REPLACE Amt WITH mamt 
 
         @ line,4 SAY mpayfrom 
         @ line,32 SAY mamt PICTURE "9999999.99" 
 
         IF again 
            * if second time through, skip to next deposit 
            SKIP 
            IF EOF() 
               EXIT 
            ENDIF 
         ENDIF 
 
         * increment line counter by one 
         line = line + 1 
 
      ENDDO 
 
      IF subtotal > 0 
         EXIT 
      ENDIF 
 
      * if no deposits... 
      @ line,52 SAY "NO DEPOSITS ENTERED" 
 
      * option to abort 
      ans = " " 
      DO WHILE .NOT. ans$"YyNn" 
         ans = " " 
         @ line+1,48 SAY "Do you wish to abort? (Y/N)" GET ans 
         READ 
      ENDDO 
 
      * erase displays from 'line' and 'line+1' 
      @ line,48 CLEAR 
 
      IF UPPER(ans)="Y" 
         @ line,55 SAY "PROGRAM ABORTED" 
         @ line+1,52 SAY "Returning to main menu" 



         PACK 
         * close out files and return to main menu 
         CLOSE DATABASES 
         RETURN 
      ELSE 
         * if second time through loop 
         IF again 
            PACK 
            restart = .T. 
            lisa = .F. 
         ENDIF 
 
      ENDIF 
 
   ENDDO lisa 
 
   IF restart 
      STORE .F. TO special,again 
      LOOP 
   ENDIF 
 
   @ 18,30 SAY LEFT(shoriz,13) 
   @ 17,15 SAY "LESS CHANGE" 
 
   * input cash withdrawal - compare to subtotal 
   totals = 0.00 
   DO WHILE .T. 
      @ 17,32 GET cash PICTURE "9999999.99" 
      READ 
      totals = subtotal - cash 
      IF totals > 0.00 
         EXIT 
      ENDIF 
      @ 22,10 SAY "Cash withdrawal exceeds deposit - please reenter" 
   ENDDO 
   @ 22,0 
 
   @ 17,32 SAY cash PICTURE "9999999.99" 
 
   * put brackets around amount if not zero 
   IF cash <> 0 
      @ 17,31 SAY "(" 
      @ 17,42 SAY ")" 
   ENDIF 
 
   @ 19,15 SAY "TOTAL DEPOSIT" 
   @ 19,32 SAY totals PICTURE "9999999.99" 
 
   * check for correctness 
   @ 22,6 SAY "Is the above deposit slip correct? "+; 
              "[ (Y)es / (N)o / (A)bandon ]" 
   answer = " " 
   DO WHILE .NOT. answer$"YyNnAa" 
      answer = " " 



      @ 22,70 GET answer 
      READ 
   ENDDO 
 
   IF UPPER(answer)="N" 
      GOTO firstdep 
      again = .T. 
      LOOP 
   ENDIF 
 
   IF UPPER(answer)="A" 
      @ 22,0 
      @ 22,12 SAY "ABORTING ABOVE DEPOSIT" 
      GOTO firstdep 
      DELETE REST 
      PACK 
   ENDIF 
 
   IF UPPER(answer)="Y" 
      balance = balance + totals 
 
      IF cash > 0 
         APPEND BLANK 
         REPLACE Date WITH mdate 
         REPLACE Amt WITH -cash 
         REPLACE Paidfrom WITH "Cash Withdrawal" 
      ENDIF 
 
      IF again 
         PACK 
      ENDIF 
 
      lastdep = lastdep + 1 
 
      USE Bank    
      APPEND BLANK 
      REPLACE Date WITH mdate 
      REPLACE Amt WITH totals 
      REPLACE Num WITH lastdep 
      REPLACE Clear WITH .F. 
      USE Deposits 
   ENDIF 
 
   * Would you like to enter another deposit? 
   answer = " " 
   DO WHILE .NOT. answer$"YyNn" 
      answer = " " 
      @ 22,0 
      @ 22, 3 SAY "Would you like to enter another deposit? (Y/N) " 
      @ 22,50 GET answer 
      READ 
   ENDDO 
 
   IF UPPER(answer)="N" 



      @ 22,0 
      @ 22,10 SAY "Returning to the main menu" 
      EXIT 
   ELSE 
      firstdep = RECCOUNT() + 1 
      again = .F. 
   ENDIF 
 
ENDDO 
 
* close out file and return to the Main menu 
CLOSE DATABASES 
RETURN 
* Eof: Add.prg���������������������������������������� 


